
TflInterp
    

Tensorflow lite interpreter for Elixir.
Deep Learning inference framework for embedded devices.
Design policy (Features)
TflInterp is designed based on the following policy.
	Provide only the Deep Learning inference. It aims to the poor-resource devices such as IOT and mobile.
	Easy to understand. The inference part, excluding pre/post-processing, can be written in a few lines.
	Use trained models from major Deep Learning frameworks that are easy to obtain.
	Multiple inference models can be used from a single application.
	There are few dependent modules. It does not have image processing or matrix calculation functions.
	TflInterp does not block the erlang/elixir process scheduler. It runs as an OS process outside of elixir.
	The back-end inference engine can be easily replaced. It's easy to keep up with the latest Deep Learninig technology.

And I'm trying to make TflInterp easy to install. 
short or concise history
The development of Tflinterp started in 2020 Nov. The original idea was to use Nerves to create an AI remote controlled car.
In the first version, I implemented Yolo3, but the design strongly depended on that model, which made it difficult to use in other applications.
Reflecting on that mistake, I redesigned Tflinterp according to the above design guidelines.
Platform
It has been confirmed to work in the following OS environment.
	Windows Visual C++ 2019
	WSL2/Ubuntu 20.04
	Nerves ARMv6, ARMv7NEON and AArch64

Requirements
	cmake 3.18.6 or later
	git
	Visual C++ 2019 for Windows

Installation
Since 0.1.3, the installation method of this module has changed.
You may need to remove previously installed TflInterp before installing new version.
There are two installation methods. You can choose either one according to your purpose.
Method-1. Like any other elixir module, add TflInterp to the dependency list in the mix.exs.
def deps do
  [
    ...
    {:tfl_interp, "~> 0.1.11"},
  ]
end
Method-2. Download TflInterp to a directory and build it ahead of time.
# download TflInterp in advance.
$ cd /home/{your home}/workdir
$ git clone -b nerves https://github.com/shoz-f/tfl_interp.git
$ cd tfl_interp
$ mix deps.get
$ mix compile
After adding the directory path to the dependency list in your mix.exs file, add the following line to the deps() function: System.put_env("SKIP_MAKE_TFLINTERP", "YES")
This line sets an environment variable that instructs your TflInterp application to use the precompiled tfl_interp.exe file, eliminating the need to build it again.
def deps do
  System.put_env("SKIP_MAKE_TFLINTERP", "YES)
  [
    ...
    {:tfl_interp, path: "/home/{your home}/workdir/tfl_interp"}
  ]
end
Then you run the following commands in your application project.
For Desktop application:
$ mix deps.get
$ mix compile

For Nerves application:
$ export MIX_TARGET=rpi3  # <- specify target device tag
$ mix deps.get
$ mix firmware

It takes a long time to finish the build. Because it will download the required files - Tensorflow sources,
ARM toolchain [^1], etc - at the first build time.
Method 1 saves the downloaded files under "{your app}/deps/tfl_interp". On the other hand,
method 2 saves them under "/home/{your home}/workdir/tfl_interp".
If you want to reuse the downloaded files in other applications, I recommend Method 2.
In either method 1 or 2, the external modules required for Tensorflow lite are stored under
their "_build/{target}/.cmake_build" according to the cmakelists.txt that comes with Tensorflow.
 [^1] Unfortunately, the ARM toolchain that comes with Nerves can not build Tensorflow lite. We need to get the toolchain recommended by the Tensorflow project.
After installation, you will have the directory tree like these:
Method 1
work_dir
  +- your-app
       +- _build/
       |    +- dev/
       |         +- .cmake_build/ --- CMakeCash.txt and external modules that Tensorflowlite depends on.
       |         |                    The cmake build outputs are stored here also.
       |         +- lib/
       |         |    +- tfl_interp
       |         |         +- ebin/
       |         |         +- priv
       |         |              +- tfl_interp --- executable: tensorflow interpreter.
       |         :
       |
       +- deps/
       |    + tfl_interp
       |    |   +- 3rd_party/ --- Tensorflow sources, etc.
       |    |   +- lib/ --- TflInterp module.
       |    |   +- src/ --- tfl_interp C++ sources.
       |    |   +- test/
       |    |   +- toolchain/ --- ARM toolchains for Nerves.
       |    |   +- CMakeLists.txt --- CMake configuration for for building tfl_interp.
       |    |   +- mix.exs
       |    :
       |
       +- lib/
       +- test/
       +- mix.exs
Method 2
work_dir
  +- your-app
  |    +- _build/
  |    |    +- dev/
  |    |         +- lib/
  |    |         |    +- tfl_interp
  |    |         |         +- ebin/
  |    |         |         +- priv
  |    |         |              +- tfl_interp --- executable: tensorflow interpreter.
  |    |         |                                copy it from the tfl_interp project.
  |    |         :
  |    |
  |    +- deps/
  |    +- lib/
  |    +- test/
  |    +- mix.exs
  |
  +- tfl_interp
       +- 3rd_party/ --- Tensorflow sources, etc.
       +- _build/
       |    +- dev/
       |         +- .cmake_build/ --- CMakeCash.txt and external modules that Tensorflowlite depends on.
       |         |                    The cmake build outputs are stored here also.
       |         +- lib/
       |         |    +- tfl_interp
       |         |         +- ebin/
       |         |         +- priv
       |         |              +- tfl_interp --- executable: tensorflow interpreter.
       |         :
       |
       +- deps/
       +- lib/ --- TflInterp module.
       +- src/ --- tfl_interp C++ sources.
       +- test/
       +- toolchain/ --- ARM toolchains for Nerves.
       +- CMakeLists.txt --- CMake configuration for for building tfl_interp.
       +- mix.exs
Basic Usage
You get the trained tflite model and save it in a directory that your application can read.
"your-app/priv" may be good choice.
$ cp your-trained-model.tflite ./priv

Next, you will create a module that interfaces with the deep learning model. The module will need pre-processing and
post-processing in addition to inference processing, as in the example following. TflInterp provides inference processing only.
You put use TflInterp at the beginning of your module, specify the model path as an optional argument. In the inference
section, you will put data input to the model (TflInterp.set_input_tensor/3), inference execution (TflInterp.invoke/1),
and inference result retrieval (TflInterp.get_output_tensor/2).
defmodule YourApp.YourModel do
  use TflInterp, model: "priv/your-trained-model.tflite"

  def predict(data) do
    # preprocess
    #  to convert the data to be inferred to the input format of the model.
    input_bin = convert-float32-binaries(data)

    # inference
    #  typical I/O data for Tensorflow lite models is a serialized 32-bit float tensor.
    output_bin =
      __MODULE__
      |> TflInterp.set_input_tensor(0, input_bin)
      |> TflInterp.invoke()
      |> TflInterp.get_output_tensor(0)

    # postprocess
    #  add your post-processing here.
    #  you may need to reshape output_bin to tensor at first.
    tensor = output_bin
      |> Nx.from_binary({:f, 32})
      |> Nx.reshape({size-x, size-y, :auto})

    * your-postprocessing *
    ...
  end
end
Let's enjoy ;-)
Demo
There is MNIST web application in demo_mnist directory. You can do it by following the steps below.
$ cd demo_mnist
$ mix deps.get
$ mix run --no-halt

And then, please open your browser with "http://localhost:5000".
License
TflInterp is licensed under the Apache License Version 2.0.

Changelog
    

Release 0.1.13(Jun 4 2023)
	Major Features and Improvements	(experimental) updated backend loader.



Release 0.1.12(May 31 2023)
	Major Features and Improvements	(experimental) replace environment variable SKIP_MAKE_TFLINTERP with "NNCOMPILED" to use the precompiled
tfl_interp.exe. now, the precompiled tfl_interp.exe is downloaded from GitHub at runtime.



Release 0.1.11(Apl 29 2023)
	Major Features and Improvements
	included custom operations: ExtractImagePatches.


	Bug Fixes and Other Changes
	add demo applications: Candy, Midas, YoloX and DeepFillV2.
	revised to shallow clone Tensorflow file set.



Release 0.1.10(Apl 12 2023)
	Major Features and Improvements
	TensorFlow background process has been updated to version 2.12.
	model downloader now includes a progress bar feature.
	environment variable SKIP_MAKE_TFLINTERP to use the precompiled tfl_interp.exe.
	included custom operations: MaxPoolingWithArgmax2D, MaxUnpooling2D, Convolution2DTransposeBias.
	added demo_hairsegmentation.


	Bug Fixes and Other Changes
	the build tool chain in Windows has been replaced from MINGW gcc to Visual C++ 2019.



Release 0.1.9(Dec 18 2022)
	Major Features and Improvements	feed back from AxonInterp project.
	added demo_yolov4.
	revised documents.
	added the ability to download model files from the url.
	added the ability to give hints about I/O tensor of the model.



Release 0.1.8(Aug 15 2022)
	Breaking Changes (from OnnxInterp 0.1.5)
	the previous download source of nlohmann-json is gone, so changed to another one.


	Bug Fixes and Other Changes (from OnnxInterp 0.1.4)
	added two more ways to specify box in NMS.



Release 0.1.7(Jul 11 2022)
* fixed non_max_suppression_multi_class/7.
Release 0.1.6(Jun 29 2022)
	Major Features and Improvements	Feedback based on OnnxInterp design.
	update Tensorflow to 2.9.1.



Release 0.1.5(Apl 4 2022)
	Major Features and Improvements	add session mode.
	add processing time measurement function.
	update Tensorflow to 2.8.0.



Release 0.1.4(Feb 6 2022)
	Bug Fixes and Other Changes	adjust word alignment of NMS parameter to ARMs.



Release 0.1.3(Jan 30 2022)
	Breaking Changes
	change installation methods. see README.md for detail.


	Major Features and Improvements
	move the directory to download Tensorflow to "3rd_party".
	supported Nerves ARMv6, ARMv7NEON and AArch64(not yet tested).


	Bug Fixes and Other Changes
	corrected CMakeLists scripts for MSYS2/MinGW64. 
	revised demo_mnist to fit cimg_ex 0.1.8.



Release 0.1.2(Jan 17 2022)
	Breaking Changes

	Major Features and Improvements
	remove C++ code that depend on C++20 features: src/tfl_interp.cc.
	limit dependency of mix_cmake to dev.


	Bug Fixes and Other Changes

Pose Estimation: MoveNet
    

File.cd!(__DIR__)
# for windows JP
System.shell("chcp 65001")
System.put_env("SKIP_MAKE_TFLINTERP", "YES")

Mix.install([
  {:tfl_interp, path: ".."},
  {:nx, "~> 0.4.0"},
  {:cimg, "~> 0.1.15"},
  {:exprintf, "~> 0.2.1"},
  {:kino, "~> 0.7.0"}
])
0.Original work
	MoveNet: Ultra fast and accurate pose detection model<br>
https://www.tensorflow.org/hub/tutorials/movenet

Thanks a lot!!!

Implementation with TflInterp in Elixir
1.Defining the inference module: MoveNet
	Model
lite-model_movenet_singlepose_lightning_tflite_int8_4.tflite

	Pre-processing
Resize the input image to the size {192,192}, 32bit integer to {0,255} and keep aspect ratio.

	Post-processing
output[0] - 17 keypoints (in the order of: [nose, left eye, right eye, left ear, right ear, left shoulder, right shoulder, left elbow, right elbow, left wrist, right wrist, left hip, right hip, left knee, right knee, left ankle, right ankle]). Make pairs of keypoints as "bone"


defmodule Movenet do
  @width 192
  @height 192

  alias TflInterp, as: NNInterp

  use NNInterp,
    model: "./model/lite-model_movenet_singlepose_lightning_tflite_int8_4.tflite",
    url:
      "https://tfhub.dev/google/lite-model/movenet/singlepose/lightning/tflite/int8/4?lite-format=tflite",
    inputs: [u8: {1, @height, @width, 3}],
    outputs: [f32: {1, 1, 17, 3}]

  @prepro CImg.builder()
          |> CImg.resize({@width, @height}, :ul, 0)
          |> CImg.to_binary(dtype: "<u1")

  def apply(img) do
    # preprocess
    input0 = CImg.run(@prepro, img)

    # prediction
    output0 =
      session()
      |> NNInterp.set_input_tensor(0, input0)
      |> NNInterp.invoke()
      |> NNInterp.get_output_tensor(0)
      |> Nx.from_binary(:f32)
      |> Nx.reshape({17, 3})

    # postprocess
    {inv_w, inv_h} = inv_aspect(img)
    joints = Nx.multiply(output0, Nx.tensor([inv_h, inv_w, 1.0]))

    {:ok, to_bones(joints)}
  end

  @bones [
    {0, 1, :fuchsia},
    {0, 2, :aqua},
    {1, 3, :fuchsia},
    {2, 4, :aqua},
    {0, 5, :fuchsia},
    {0, 6, :aqua},
    {5, 7, :fuchsia},
    {7, 9, :fuchsia},
    {6, 8, :aqua},
    {8, 10, :aqua},
    {5, 6, :yellow},
    {5, 11, :fuchsia},
    {6, 12, :aqua},
    {11, 12, :yellow},
    {11, 13, :fuchsia},
    {13, 15, :fuchsia},
    {12, 14, :aqua},
    {14, 16, :aqua}
  ]

  def to_bones(t, threshold \\ 0.11) do
    Enum.flat_map(@bones, fn {p1, p2, color} ->
      [y1, x1, score1] = Nx.to_flat_list(t[p1])
      [y2, x2, score2] = Nx.to_flat_list(t[p2])

      if score1 > threshold && score2 > threshold do
        [{x1, y1, x2, y2, color}]
      else
        []
      end
    end)
  end

  defp inv_aspect(img) do
    {w, h, _, _} = CImg.shape(img)
    if w > h, do: {1.0, w / h}, else: {h / w, 1.0}
  end
end
Launch Movenet.
# TflInterp.stop(Movenet)
Movenet.start_link([])
Display the properties of the Movenet model.
TflInterp.info(Movenet)
2.Defining execution module DemoMovenet
defmodule LiveMovenet do
  def img_list(src, range) do
    Enum.map(range, fn i ->
      name = ExPrintf.sprintf("%03d.jpg", [i])
      path = Path.join(src, name)
      CImg.load(path)
    end)
  end

  def animate(img_list) do
    Kino.animate(img_list, fn img ->
      draw_movenet(img)
      |> CImg.display_kino(:jpeg)
    end)
  end

  def run(path) do
    CImg.load(path)
    |> draw_movenet()
    |> CImg.display_kino(:jpeg)
  end

  defp draw_movenet(img) do
    with {:ok, res} <- Movenet.apply(img) do
      Enum.reduce(res, CImg.builder(img), fn {x1, y1, x2, y2, color}, canvas ->
        CImg.draw_line(canvas, x1, y1, x2, y2, color, thick: 5)
      end)
    end
  end
end
3.Let's try it
img_list = LiveMovenet.img_list("young-people-dancing", 1..200)
LiveMovenet.animate(img_list)
4.TIL ;-)
&#9633;

Object Detection: YOLOv4
    

File.cd!(__DIR__)
# for windows JP
# System.shell("chcp 65001")
System.put_env("SKIP_MAKE_TFLINTERP", "YES")

Mix.install([
  {:tfl_interp, path: ".."},
  {:cimg, "~> 0.1.14"},
  {:postdnn, "~> 0.1.5"},
  {:kino, "~> 0.7.0"}
])
0.Original work
Pytorch-YOLOv4
	https://github.com/Tianxiaomo/pytorch-YOLOv4

This note usee the pretraind model converted from above project ;-)
Thanks a lot!!!

Implementation with TflInterp in Elixir
1.Defining the inference module: YOLOv4
	Model
yolov4_1_3_608_608_static.tflite

	Pre-processing
Resize the input image to the size {608,608}, normalize to {0.0,1.0} and transpose NCHW.

	Post-processing
Filtering Boxes(output[0]) and Scores(output[1]) with Multi-class Non Maximum Suppression.


defmodule YOLOv4 do
  @moduledoc """
  Original work:
    Pytorch-YOLOv4 - https://github.com/Tianxiaomo/pytorch-YOLOv4
  """

  alias TflInterp, as: NNInterp

  use NNInterp,
    model: "./model/yolov4_1_3_608_608_static.tflite",
    url:
      "https://drive.google.com/uc?authuser=0&export=download&confirm=t&id=1z02aAUJ_phk3mom_w7J1fKbe9F_UM1HT",
    inputs: [f32: {1, 3, 608, 608}],
    outputs: [f32: {1, 22743, 1, 4}, f32: {1, 22743, 80}]

  def apply(img) do
    # preprocess
    input0 =
      img
      |> CImg.resize({608, 608})
      |> CImg.to_binary([{:range, {0.0, 1.0}}, :nchw])

    # prediction
    outputs =
      session()
      |> NNInterp.set_input_tensor(0, input0)
      |> NNInterp.invoke()

    # postprocess
    boxes = NNInterp.get_output_tensor(outputs, 0)
    scores = NNInterp.get_output_tensor(outputs, 1)

    PostDNN.non_max_suppression_multi_class(
      {22743, 80},
      boxes,
      scores,
      boxrepr: :corner,
      label: "./model/coco.label"
    )
  end
end
Launch YOLOv4.
# OnnxInterp.stop(Resnet18)
YOLOv4.start_link([])
Display the properties of the YOLOv4 model.
TflInterp.info(YOLOv4)
2.Defining execution module LiveYOLOv4
defmodule LiveYOLOv4 do
  @palette CImg.Util.rand_palette("./model/coco.label")

  def run(path) do
    img = CImg.load(path)

    with {:ok, res} <- YOLOv4.apply(img) do
      IO.inspect(res)

      Enum.reduce(res, CImg.builder(img), &draw_item(&1, &2))
      |> CImg.display_kino(:jpeg)
    end
  end

  defp draw_item({name, boxes}, canvas) do
    color = @palette[name]

    Enum.reduce(boxes, canvas, fn [_score, x1, y1, x2, y2, _index], canvas ->
      [x1, y1, x2, y2] = PostDNN.clamp([x1, y1, x2, y2], {0.0, 1.0})

      CImg.fill_rect(canvas, x1, y1, x2, y2, color, 0.35)
    end)
  end
end
3.Let's try it
LiveYOLOv4.run("dog.jpg")
Appendix
&#9633;

Resnet18 image classification
    

File.cd!(__DIR__)
# for windows JP
# System.shell("chcp 65001")
System.put_env("SKIP_MAKE_TFLINTERP", "YES")

Mix.install([
  {:tfl_interp, path: ".."},
  {:cimg, "~> 0.1.14"},
  {:nx, "~> 0.4.0"},
  {:kino, "~> 0.7.0"}
])
0.Original work
torchvision.models.resnet18 -  pre-trained model included in Pytorch.
	https://pytorch.org/vision/main/models/generated/torchvision.models.resnet18.html

Thanks a lot!!!

Implementation with TflInterp in Elixir
1.Defining the inference module: ResNet18
	Model
resnet18.axon: get from "https://github.com/shoz-f/tfl_interp/releases/download/0.0.1/resnet18.tflite" if not existed.

	Pre-processing
Resize the input image to the size {@width, @height} and gaussian normalize.

	Post-processing
Sort outputs and take first item.


defmodule Resnet18 do
  @width 224
  @height 224

  alias TflInterp, as: NNInterp

  use NNInterp,
    model: "./model/resnet18.tflite",
    url: "https://github.com/shoz-f/tfl_interp/releases/download/0.0.1/resnet18.tflite",
    inputs: [f4: {1, 3, @height, @width}],
    outputs: [f4: {1, 1000}]

  @imagenet1000 (for item <- File.stream!("./imagenet1000.label") do
                   String.trim_trailing(item)
                 end)
                |> Enum.with_index(&{&2, &1})
                |> Enum.into(%{})

  def apply(img, top \\ 1) do
    # preprocess
    input0 =
      CImg.builder(img)
      |> CImg.resize({@width, @height})
      |> CImg.to_binary([{:gauss, {{123.7, 58.4}, {116.3, 57.1}, {103.5, 57.4}}}, :nchw])

    # prediction
    output0 =
      session()
      |> NNInterp.set_input_tensor(0, input0)
      |> NNInterp.invoke()
      |> NNInterp.get_output_tensor(0)
      |> Nx.from_binary(:f32)
      |> Nx.reshape({1000})

    # postprocess
    # softmax
    then(Nx.exp(output0), fn exp -> Nx.divide(exp, Nx.sum(exp)) end)
    |> Nx.argsort(direction: :desc)
    |> Nx.slice([0], [top])
    |> Nx.to_flat_list()
    |> Enum.map(&@imagenet1000[&1])
  end
end
Launch Resnet18.
# TflInterp.stop(Resnet18)
Resnet18.start_link([])
Display the properties of the Resnet18 model.
TflInterp.info(Resnet18)
2.Let's try it
Load a photo and apply Resnet18 to it.
img = CImg.load("lion.jpg")
Kino.render(CImg.display_kino(img, :jpeg))

Resnet18.apply(img, 3)
&#9633;

Hair Segmentation
    

File.cd!(__DIR__)
# for windows JP
System.shell("chcp 65001")
System.put_env("SKIP_MAKE_TFLINTERP", "YES")

Mix.install([
  {:tfl_interp, path: ".."},
  {:cimg, "~> 0.1.19"},
  {:nx, "~> 0.4.0"},
  {:kino, "~> 0.7.0"}
])
0.Original work
"Real-time hair segmentation and recoloring on Mobile GPUs", Google Research.
	https://sites.google.com/view/perception-cv4arvr/hair-segmentation

Thanks a lot!!!

Implementation with TflInterp in Elixir
1.Defining the inference module: HairSegmentation
	Model
hair_segmentation.tflite: get from "https://storage.googleapis.com/mediapipe-assets/hair_segmentation.tflite" if not existed.

	Pre-processing
Resize the input image to the size {@width, @height} and normalize {0.0, 1.0}.

	Post-processing
Overlap the mask on the input image.


defmodule HairSegmentation do
  @width 512
  @height 512

  alias TflInterp, as: NNInterp

  use NNInterp,
    model: "./model/hair_segmentation.tflite",
    url: "https://storage.googleapis.com/mediapipe-assets/hair_segmentation.tflite",
    inputs: [f32: {1, @width, @height, 4}],
    outputs: [f32: {1, @width, @height, 2}]

  def apply(img) do
    # preprocess
    input0 =
      CImg.builder(img)
      |> CImg.resize({@width, @height})
      |> CImg.append(CImg.create(@width, @height, 1, 1, 0), :c)
      |> CImg.to_binary([{:range, {0.0, 1.0}}])

    # prediction
    output =
      session()
      |> NNInterp.set_input_tensor(0, input0)
      |> NNInterp.invoke()
      |> NNInterp.get_output_tensor(0)
      |> Nx.from_binary(:f32)
      |> Nx.reshape({@height, @width, :auto})

    # postprocess
    [background, hair] =
      Enum.map(0..1, fn i ->
        Nx.slice_along_axis(output, i, 1, axis: 2) |> Nx.squeeze()
      end)

    {w, h, _, _} = CImg.shape(img)

    Nx.greater(hair, background)
    |> Nx.to_binary()
    |> CImg.from_binary(@width, @height, 1, 1, dtype: "<u1")
    |> CImg.resize({w, h})  # make image
  end

  def coloring(img, color, opacity \\ 0.5) do
    mask = HairSegmentation.apply(img)
    CImg.paint_mask(img, mask, color, opacity)
  end
end
Launch HairSegmentation.
# TflInterp.stop(HairSegmentation)
HairSegmentation.start_link([])
Display the properties of the HairSegmentation model.
TflInterp.info(HairSegmentation)
2.Let's try it
Load a photo and apply HairSegmentation to it.
img = CImg.load("photo_girl.jpg")
colored = HairSegmentation.coloring(img, [{0, 255, 0}], 0.3)

Enum.map([img, colored], &CImg.display_kino(&1, :jpeg))
|> Kino.Layout.grid(columns: 2)
&#9633;

Fast Neural Style Transfer
    

File.cd!(__DIR__)
# for windows JP
System.shell("chcp 65001")
System.put_env("SKIP_MAKE_TFLINTERP", "YES")

Mix.install([
  {:tfl_interp, path: ".."},
  {:cimg, "~> 0.1.20"},
  {:kino, "~> 0.7.0"}
])
0.Original work
Fast Neural Style Transfer
	https://github.com/onnx/models/tree/main/vision/style_transfer/fast_neural_style

fast-neural-style in PyTorch
	https://github.com/pytorch/examples/tree/main/fast_neural_style

Thanks a lot!!!

Implementation with TflInterp in Elixir
1.Defining the inference module: Candy
	Model
candy.tflite: get from "https://github.com/shoz-f/tinyML_livebook/releases/download/model/candy.tflite" if not existed.

	Pre-processing
Resize the input image to {@width, @height}, normalize it to a range of {0.0, 255.0} and transpose NCHW.

	Post-processing
The candy colored image is outputted directly by this model.


defmodule Candy do
  @width 224
  @height 224

  alias TflInterp, as: NNInterp

  use NNInterp,
    model: "./model/candy.tflite",
    url: "https://github.com/shoz-f/tinyML_livebook/releases/download/model/candy.tflite",
    inputs: [f32: {1, @height, @width, 3}],
    outputs: [f32: {1, @height, @width, 3}]

  def apply(img) do
    # preprocess
    input0 =
      CImg.builder(img)
      |> CImg.resize({@width, @height})
      |> CImg.to_binary([{:range, {0.0, 255.0}}, :nchw])

    # prediction
    output =
      session()
      |> NNInterp.set_input_tensor(0, input0)
      |> NNInterp.invoke()
      |> NNInterp.get_output_tensor(0)

    # postprocess
    output
    |> CImg.from_binary(@width, @height, 1, 3, [{:range, {0.0, 255.0}}, :nchw])
    |> CImg.resize(img)
  end
end
Launch Candy.
# TflInterp.stop(Candy)
Candy.start_link([])
Display the properties of the Candy model.
TflInterp.info(Candy)
2.Let's try it
Load a photo and apply Candy to it.
img = CImg.load("flog.jpg")

result = Candy.apply(img)

Enum.map([img, result], &CImg.display_kino(&1, :jpeg))
|> Kino.Layout.grid(columns: 2)

Monocular Depth Estimation: MiDaS v2.1
    

File.cd!(__DIR__)
# for windows JP
System.shell("chcp 65001")
System.put_env("SKIP_MAKE_TFLINTERP", "YES")

Mix.install([
  {:tfl_interp, path: ".."},
  {:cimg, "~> 0.1.19"},
  {:nx, "~> 0.4.0"},
  {:kino, "~> 0.7.0"}
])
0.Original work
Intelligent Systems Lab Org:
"Towards Robust Monocular Depth Estimation: Mixing Datasets for Zero-shot Cross-dataset Transfer"
	https://arxiv.org/abs/1907.01341v3
	https://github.com/isl-org/MiDaS

Thanks a lot!!!

Implementation with TflInterp in Elixir
1.Defining the inference module: Midas
	Model
midas.tflite: get from "https://github.com/isl-org/MiDaS/releases/download/v2_1/model_opt.tflite" if not existed.

	Pre-processing
Resize input image to {@width, @height} and normalize it to a range of {-1.0, 1.0}.

	Post-processing
The depth image {@width, @height, 1} is outputted directly by this model.


defmodule Midas do
  @width 256
  @height 256

  alias TflInterp, as: NNInterp

  use NNInterp,
    model: "./model/model_opt.tflite",
    url: "https://github.com/isl-org/MiDaS/releases/download/v2_1/model_opt.tflite",
    inputs: [f32: {1, @height, @width, 3}],
    outputs: [f32: {1, @height, @width, 3}]

  def apply(img) do
    # preprocess
    input0 =
      CImg.builder(img)
      |> CImg.resize({@width, @height})
      |> CImg.to_binary(range: {-1.0, 1.0})

    # prediction
    output =
      session()
      |> NNInterp.set_input_tensor(0, input0)
      |> NNInterp.invoke()
      |> NNInterp.get_output_tensor(0)
      |> Nx.from_binary(:f32)
      |> Nx.reshape({@height, @width})

    # postprocess
    [min, max] =
      [Nx.window_min(output, {@height, @width}), Nx.window_max(output, {@height, @width})]
      |> Enum.map(&Nx.squeeze/1)
      |> Enum.map(&Nx.to_number/1)

    {w, h, _, _} = CImg.shape(img)

    Nx.subtract(output, min)
    |> Nx.divide(max - min)
    |> Nx.to_binary()
    |> CImg.from_binary(@width, @height, 1, 1)
    |> CImg.resize({w, h})
  end
end
Launch Midas.
# TflInterp.stop(Midas)
Midas.start_link([])
Display the properties of the Midas model.
TflInterp.info(Midas)
2.Let's try it
Load a photo and apply Midas to it.
img = CImg.load("sample.jpg")

result =
  Midas.apply(img)
  |> CImg.color_mapping(:jet)

Enum.map([img, result], &CImg.display_kino(&1, :jpeg))
|> Kino.Layout.grid(columns: 2)
&#9633;

Object Detection: YoloX
    

File.cd!(__DIR__)
# for windows JP
System.shell("chcp 65001")
System.put_env("SKIP_MAKE_TFLINTERP", "YES")

Mix.install([
  {:tfl_interp, path: ".."},
  {:cimg, "~> 0.1.19"},
  {:nx, "~> 0.4.0"},
  {:postdnn, "~> 0.1.5"},
  {:kino, "~> 0.7.0"}
])
0.Original work
Zheng Ge, Zheng and Liu, Songtao and Wang, Feng and Li, Zeming and Sun, Jian
"YOLOX: Exceeding YOLO Series in 2021"
	https://arxiv.org/abs/2107.08430
	https://github.com/Megvii-BaseDetection/YOLOX

A technical article on YOLOX in Japanese
@koshian2 "実装から見るYOLOX：2021年のYOLOシリーズを超えて"
	https://qiita.com/koshian2/items/af032cb102f48e789e66


Thanks a lot!!!

Implementation with TflInterp in Elixir
1.Defining the inference module: YoloX
	Model
yolox.tflite: get from "https://github.com/shoz-f/tinyML_livebook/releases/download/model/yolox_s.tflite" if not existed.

	Pre-processing
Resize input image to {@width, @height}, normalize it to a range of {0.0, 255.0} and transpose NCHW.

	Post-processing
Divide the output to boxes and scores, and then filter them with Multi-class Non Maximum Suppression.


defmodule YoloX do
  @width 640
  @height 640

  alias TflInterp, as: NNInterp

  use NNInterp,
    model: "./model/yolox_s.tflite",
    label: "./model/coco.label",
    url: "https://github.com/shoz-f/tinyML_livebook/releases/download/model/yolox_s.tflite",
    inputs: [f32: {1, 3, @height, @width}],
    outputs: [f32: {1, 8400, 85}]

  def apply(img) do
    # preprocess
    input0 =
      CImg.builder(img)
      |> CImg.resize({@width, @height})
      |> CImg.to_binary([{:range, {0.0, 255.0}}, :nchw])

    # prediction
    output =
      session()
      |> NNInterp.set_input_tensor(0, input0)
      |> NNInterp.invoke()
      |> NNInterp.get_output_tensor(0)
      |> Nx.from_binary(:f32)
      |> Nx.reshape({:auto, 85})

    # postprocess
    output = Nx.transpose(output)

    boxes = extract_boxes(output)
    scores = extract_scores(output)

    TflInterp.non_max_suppression_multi_class(
      __MODULE__,
      Nx.shape(scores),
      Nx.to_binary(boxes),
      Nx.to_binary(scores)
    )
  end

  @grid PostDNN.meshgrid({@width, @height}, [8, 16, 32], [:transpose, :normalize])

  defp extract_boxes(t) do
    # decode box center coordinate on {1.0, 1.0}
    center =
      t[0..1]
      # * pitch(x,y)
      |> Nx.multiply(@grid[2..3])
      # + grid(x,y)
      |> Nx.add(@grid[0..1])

    # decode box size
    size =
      t[2..3]
      |> Nx.exp()
      # * pitch(x,y)
      |> Nx.multiply(@grid[2..3])

    Nx.concatenate([center, size]) |> Nx.transpose()
  end

  defp extract_scores(t) do
    Nx.multiply(t[4], t[5..-1//1])
    |> Nx.transpose()
  end
end
Launch YoloX.
# TflInterp.stop(YoloX)
YoloX.start_link([])
Display the properties of the YoloX model.
TflInterp.info(YoloX)
2.Defining execution module DemoYoloX
defmodule DemoYoloX do
  @palette CImg.Util.rand_palette("./model/coco.label")

  def run(path) do
    img = CImg.load(path)

    with {:ok, res} <- YoloX.apply(img) do
      IO.inspect(res)

      Enum.reduce(res, CImg.builder(img), &draw_item(&1, &2))
      |> CImg.display_kino(:jpeg)
    end
  end

  defp draw_item({name, boxes}, canvas) do
    color = @palette[name]

    Enum.reduce(boxes, canvas, fn [_score, x1, y1, x2, y2, _index], canvas ->
      [x1, y1, x2, y2] = PostDNN.clamp([x1, y1, x2, y2], {0.0, 1.0})

      CImg.fill_rect(canvas, x1, y1, x2, y2, color, 0.35)
    end)
  end
end
3.Let's try it
Load a photo and apply YoloX to it.
DemoYoloX.run("dog.jpg")
&#9633;

Generative Inpainting
    

File.cd!(__DIR__)
# for windows JP
System.shell("chcp 65001")
System.put_env("SKIP_MAKE_TFLINTERP", "YES")

Mix.install([
  {:tfl_interp, path: ".."},
  {:cimg, "~> 0.1.19"},
  {:kino, "~> 0.7.0"}
])
0.Original work
"Generative Image Inpainting with Contextual Attention"
	https://arxiv.org/abs/1801.07892

"Free-Form Image Inpainting with Gated Convolution"
	https://arxiv.org/abs/1806.03589

GitHub: Generative Image Inpainting
	https://github.com/JiahuiYu/generative_inpainting

The tflite model deepfillv2.tflite is converted from their pretraind model.
Thanks a lot!!!

Implementation with TflInterp in Elixir
1.Defining the inference module: DeepFillV2
	Model
deepfillv2.tflite: get from "https://github.com/shoz-f/tfl_interp/releases/download/0.0.1/deepfillv2.tflite" if not existed.

	Pre-processing
Combine the original and mask images into a single image, then resize it to {@width, @height} and normalize it to a range of {0.0, 255.0} for further manipulation.

	Post-processing
The inpainted image is outputted directly by this model.


defmodule DeepFillV2 do
  @width 680
  @height 512

  alias TflInterp, as: NNInterp

  use NNInterp,
    model: "./model/deepfillv2.tflite",
    url: "https://github.com/shoz-f/tfl_interp/releases/download/0.0.1/deepfillv2.tflite",
    inputs: [f32: {1, @height, 2 * @width, 3}],
    outputs: [u8: {1, @height, @width, 3}]

  def apply(img, mask) do
    # preprocess
    input0 =
      CImg.builder(img)
      |> CImg.append(mask, :x)
      |> CImg.resize({2 * @width, @height})
      |> CImg.to_binary(range: {0.0, 255.0})

    # prediction
    {w, h, _, _} = CImg.shape(img)

    session()
    |> NNInterp.set_input_tensor(0, input0)
    |> NNInterp.invoke()
    |> NNInterp.get_output_tensor(0)
    |> CImg.from_binary(@width, @height, 1, 3, [{:dtype, "<u1"}, :bgr])
    |> CImg.resize({w, h})
  end
end
Launch DeepFillV2.
# TflInterp.stop(DeepFillV2)
DeepFillV2.start_link([])
Display the properties of the DeepFillV2 model.
TflInterp.info(DeepFillV2)
2.Let's try it
Load a photo and apply DeepFillV2 to it.
origin = CImg.load("sample_raw.jpg")
mask = CImg.load("sample_mask.jpg")

result = DeepFillV2.apply(origin, mask)

Enum.map([origin, mask, result], &CImg.display_kino(&1, :jpeg))
|> Kino.Layout.grid(columns: 2)
&#9633;

